/* Humidity Chamber Control Program V2.1 
 Written by: Robert Walker, 6/18/2013
 
 This program uses the Arduino Uno to monitor humidity levels in a test chamber using a DHT22 
 humidity/temperature sensor. If humidity levels in the chamber drop below a specified level, 
 the Arduino microcontroler turns on a humidifier located in a seperate chamber, moist air is 
 then pumped to the test chamber.  When the humidity reaches a specified shut off level, the 
 air pump and humidifier are then turned off. This cycle can be repeated until the testing is 
 completed.  
 
 V2.1 - Corrected control loops, utilizes improved circuitry.
 V2.0 - Never used, control loops were improved, reduced # of LEDs 
 V1.0 - Initial humidity control program, works, but rough control loops 
 
 DHT22 Sensor:
 Looking at the front of the sensor:
 
 Connect pin 1 (on the left) of the sensor to +5V
 Connect pin 2 of the sensor to a digital pin on Arduino
 pin 3 LEAVE UNCONNECTED
 Connect pin 4 (on the right) of the sensor to GROUND
 Connect a 10K resistor from pin 2 (data) to pin 1 (power) of the sensor
 
 DHT22 Sensor has a slow readout time, reading temperature or humidity takes about 
 250 milliseconds, and sensor readings may be up to 2 seconds old. 
 
 Humidity is in percent (%), Temperature is in degrees Celcius (C).
 */



#include "DHT.h"

#define DHTPIN 2                     //  Defining sensor arduino pin number 
#define DHTTYPE DHT22                //  Defining sensor type

DHT dht(DHTPIN, DHTTYPE);

int redLED = 11;
int greenLED = 10;
int humidify = 7;                    //  Humidifier
int pump = 6;                        //  Pump on LED light

float h;                             //  Humidity variable
float diff;                          //  diff = humidity reading (h) - humidity limit (HL)
float hyst = 1.5;                    //  Hysteresis threshold on/off (HL +1.5%)
float HL = 10;                       //  Humidity Limit

/*=======================SETUP=========================*/
//  Loads once.
//  Sets Arduino pins as outputs, begins Serial, Starts humidity sensor

void setup() {

  pinMode(redLED, OUTPUT);
  pinMode(greenLED, OUTPUT);
  pinMode(humidify, OUTPUT);
  pinMode(pump, OUTPUT);

  Serial.begin(9600); 

  dht.begin();
}

/*==================MAIN LOOP============================*/
//  Loops continuously
//  Measures humidity levels compares to set humidity leve
//  Controls ON/OFF of humidifier and pump depending on humidity readings

void loop() {
  
  h = dht.readHumidity();              //  FIRST h reading
  float t = dht.readTemperature();
  diff = h - HL;                       //  FIRST diff calculated
[bookmark: _GoBack]  Serial.print("Start Humidity: ");
  Serial.print(h);
  Serial.print("% \n");
  
  if(h >= HL){setColor(255,0);}
  
  if (diff >= hyst){                   //  IF h reading is HIGH
    Serial.print("First IF, humidity level good \n");
    setColor(255, 0);                  //  Geen LED ON
    delay(15000);                      //  15s delay between readings
  }


  else if (diff <= 0){                 //  IF h reading is LOW
    Serial.print("First Else If, humidity level low \n");
    setColor(0, 255);                  //  Red LED ON
    delay(1000);                       //  1s delay
    digitalWrite(humidify, HIGH);      //  Humidifier ON
    delay(2000);                       //  2s delay
    digitalWrite(pump, HIGH);          //  Air pump ON
    delay(5000);                       //  5s delay

    h = dht.readHumidity();            //  NEW h reading
    t = dht.readTemperature();
    diff = h - HL;                     //  NEW diff calculated
    Serial.print("Second Humidity Measurement: ");
    Serial.print(h);
    Serial.print("% \n");
    

    if (diff >= hyst){                 //  IF h reading is HIGH
      Serial.print("Humidity good, all off \n");
      setColor(255, 0);                //  Green LED ON
      delay(500);
      digitalWrite(humidify, LOW);     //  Humidifier OFF
      delay(250);
      digitalWrite(pump, LOW);         //  Air Pump OFF
      delay(15000);                    //  15s delay
    }


    else if (diff <= 0){                //  IF h reading is LOW
      do{                               //  Loops Amber LED flash while LOW humidity readings
        Serial.print("Looping amber light, humidity read \n");
        setColor(0, 0);                 //  LEDs OFF
        delay(1000);                    //  1s delay
        setColor(255, 255);             //  Amber LED ON
        delay(4000);                    //  4s delay

        h = dht.readHumidity();         //  NEW h reading
        t = dht.readTemperature();
        diff = h - HL;                  //  NEW diff calculated
        
         Serial.print("Amber Humidity Measurement: ");
         Serial.print(h);
         Serial.print("% \n");
        

        if (diff >= hyst){              //  IF h reading is HIGH
          Serial.print("Amber Humidity good, all off \n");
          setColor(255, 0);             //  Green LED ON
          delay(500);
          digitalWrite(humidify, LOW);  //  Humidifier OFF
          delay(250);
          digitalWrite(pump, LOW);      //  Air Pump OFF
          delay(15000);                 //  15s delay
        }
      }
      while (diff <= hyst);                //  If h reading is LOW, loop to begining of 'do'
    }   
  }
}  


void setColor(int green, int red){
  analogWrite(redLED, red);
  analogWrite(greenLED, green);
}

